
Retorna um valor ou a referência para um valor de dentro de uma tabela ou intervalo. Há duas formas da função ÍNDICE: matricial (matriz: usada para criar fórmulas únicas que produzem vários resultados ou que operam em um grupo de argumentos organizados em linhas e colunas. Um intervalo de matrizes compartilha uma fórmula comum; uma constante de matriz é um grupo de constantes usado como um argumento.) e de referência.

	Se quiser	Consulte então
	Retorna o valor de uma célula ou matriz de células especificada	Forma matricial
	Retorna a referência para células especificadas	Forma de referência

Forma matricial

Retorna o valor de um elemento em uma tabela ou matriz (matriz: usada para criar fórmulas únicas que produzem vários resultados ou que operam em um grupo de argumentos organizados em linhas e colunas. Um intervalo de matrizes compartilha uma fórmula comum; uma constante de matriz é um grupo de constantes usado como um argumento.) selecionado pelos índices de número de linha e coluna.

Use a forma de matriz se o primeiro argumento de ÍNDICE for uma constante de matriz.

ÍNDICE(matriz,núm_linha,núm_coluna)

Matriz   é um intervalo de células ou uma constante de matriz.

	
Se a matriz contiver apenas uma linha ou coluna, o argumento núm_linha ou núm_coluna correspondente é opcional.


	
Se a matriz tiver mais de uma linha e mais de uma coluna e apenas núm_linha ou núm_coluna for usado, ÍNDICE retornará uma matriz referente à linha ou coluna inteira da matriz.




Núm_linha   seleciona a linha na matriz a partir da qual um valor deverá ser retornado. Se núm_linha for omitido, núm_coluna é obrigatório.

Núm_coluna   seleciona a coluna na matriz a partir da qual um valor deverá ser retornado. Se núm_coluna for omitido, núm_linha é obrigatório.

Comentários

	Se os argumentos núm_linha e núm_coluna forem usados, ÍNDICE retornará o valor contido na célula que estiver no ponto de interseção entre núm_linha e núm_coluna.
	Se você definir núm_linha ou núm_coluna como 0 (zero), ÍNDICE retornará a matriz de valores referente à coluna ou linha inteira, respectivamente. Para usar valores retornados como uma matriz, insira a função ÍNDICE como uma fórmula de matriz (fórmula de matriz: uma fórmula que executa vários cálculos em um ou mais conjuntos de valores e retorna um único resultado ou vários resultados. As fórmulas de matriz ficam entre chaves { } e são inseridas pressionando-se CTRL+SHIFT+ENTER.) em um intervalo horizontal de células para uma linha e em um intervalo vertical de células para uma coluna. Para inserir uma fórmula de matriz, pressione CTRL+SHIFT+ENTER.
	
Núm_linha e núm_coluna devem fazer referência a uma célula dentro de uma matriz, caso contrário, ÍNDICE retornará o valor de erro #REF!.




	
	 
	1
	2
	3


	
	A	B
	Dados	Dados
	Maçãs	Limões
	Bananas	Pêras
	Fórmula	Descrição (resultado)
	=ÍNDICE(A2:B3;2;2)	O valor na interseção da segunda linha com a segunda coluna no intervalo (Pêras)
	=ÍNDICE(A2:B3;2;1)	O valor na interseção da segunda linha com a primeira coluna no intervalo (Bananas)





 

	
	 
	1
	2
	3


	
	A	B
	Fórmula	Descrição (resultado)
	=ÍNDICE({1,2;3,4};0;2)	O valor da primeira linha, segunda coluna na constante de matriz (2)
	 	O valor da segunda linha, segunda coluna na constante de matriz (4)





 Observação    A fórmula no exemplo precisa ser inserida como uma fórmula de matriz. Após copiar o exemplo em uma planilha em branco, selecione o intervalo A2:A3 iniciando pela célula da fórmula. Pressione F2 e, em seguida, pressione CTRL+SHIFT+ENTER. Se não for inserida como uma fórmula de matriz, a fórmula terá como único resultado 2.

Forma de referência

Retorna a referência da célula na interseção de linha e coluna específicas. Se a referência for formada por seleções não-adjacentes, você pode escolher a seleção que deseja observar.

ÍNDICE(ref;núm_linha;núm_coluna;núm_área)

Ref   é uma referência a um ou mais intervalos de célula.

	
Se você estiver inserindo um intervalo não adjacente para a ref, coloque ref entre parênteses.


	
Se cada área na referência contiver apenas uma linha ou coluna, o argumento núm_linha ou núm_coluna, respectivamente, será opcional. Por exemplo, para uma referência de linha única, use ÍNDICE(ref;;núm_coluna).




Núm_linha   é o número da linha em ref de onde será fornecida uma referência.

Núm_coluna   é o número da coluna em ref de onde será fornecida uma referência.

Núm_área   seleciona um intervalo em ref do qual deve ser retornada a interseção de núm_linha com núm_coluna. A primeira área selecionada ou inserida recebe o número 1, a segunda recebe o número 2, e assim por diante. Se núm_área for omitido, ÍNDICE usará a área 1.

	
Por exemplo, se ref descrever as células (A1:B4;D1:E4;G1:H4), então núm_área 1 representará o intervalo A1:B4, núm_área 2 representará o intervalo D1:E4 e núm_área 3 representará o intervalo G1:H4.




Comentários

	Depois que ref e núm_área tiverem selecionado um intervalo específico, núm_linha e núm_coluna selecionam uma célula específica: núm_linha 1 é a primeira linha do intervalo, núm_coluna 1 é a primeira coluna, e assim por diante. A referência que ÍNDICE retorna é a interseção entre núm_linha e núm_coluna.
	
Se você definiu núm_linha ou núm_coluna como 0, ÍNDICE retorna a referência para a coluna ou linha inteira, respectivamente.


	Núm_linha, núm_coluna e núm_área devem apontar para uma célula na referência; senão, ÍNDICE retornará o valor de erro #REF!. Se núm_linha e núm_coluna forem omitidos, ÍNDICE retornará a área em referência especificada por núm_área.
	O resultado da função ÍNDICE é uma referência e é interpretado como tal por outras fórmulas. Dependendo da fórmula, o valor retornado por ÍNDICE pode ser usado como uma referência ou como um valor. Por exemplo, a fórmula de macro CÉL("largura";ÍNDICE(A1:B2;1;2)) é equivalente a CÉL("largura";B1). A função CÉL usa o valor retornado por ÍNDICE como uma referência de célula. Por outro lado, uma fórmula tal como 2*ÍNDICE(A1:B2;1;2) traduz o valor retornado por ÍNDICE no número da célula B1.


	
	 
	1
	2
	3
	4
	5
	6
	7
	8
	9
	10
	11


	
	A	B	C
	Fruta	Preço	Contagem
	Maçãs	0,69	40
	Bananas	0,34	38
	Limões	0,55	15
	Laranjas	0,25	25
	Pêras	0,59	40
	 	 	 
	Amêndoas	2,80	10
	Cajus	3,55	16
	Amendoins	1,25	20
	Nozes	1,75	12
	Fórmula	Descrição (resultado)
	=ÍNDICE(A2:C6;2;3)	A interseção da segunda linha com a terceira coluna no intervalo A2:C6, que é o conteúdo da célula C3 (38)
	=ÍNDICE((A1:C6;A8:C11);2;2;2)	A interseção da segunda linha com a segunda coluna na segunda área de A8:C11, que é o conteúdo da célula B9 (3,55)
	=SOMA(ÍNDICE(A1:C11;0;3;1))	A soma da terceira coluna na primeira área do intervalo A1:C11, que é a soma de C1:C6 (216)
	=SOMA(B2:ÍNDICE(A2:C6;5;2))	A soma do intervalo que inicia em B2 e termina na interseção da quinta linha com a segunda coluna no intervalo A2:A6, que é a soma de B2:B6 (2,42)

Aprenda Excel VBA com Simplicidade de 
códigos e Eficácia, Escrevendo Menos e
Fazendo Mais.

'-------------------------------------'

+  500 Video Aulas MS Excel VBA

+  Coleção 25.000 Macros MS Excel VBA
+  141 Planilhas Instruções Loops
+  341 Planilhas WorksheetFunctions(VBA)
+    04 Módulos Como Fazer Excel VBA
+  Curso Completo MS Excel VBA
+  Planilhas Inteligentes

